

Potatoes with Contrasts

```
/* potato3.sas */
options linesize=79 noovp formdlim='_';
title 'Rotten potatoes: STA429/1007 F 2007';

data spud;
  infile 'potato2.data' firstobs=2; /* Skip the first line that R uses */
  input id bact temp rot;
  combo = 10*temp+bact; /* First digit is Temp, second is Bact */

proc means mean stddev;
  class bact temp;
  var rot;

/* Get better looking output from proc tabulate */

proc tabulate;
  class bact temp;
  var rot;
  table (temp all),(bact all) * (mean*rot);

proc glm;
  title3 'Standard 2-way ANOVA with proc glm';
  class bact temp;
  model rot=temp|bact;
  means temp*bact;

/* Now generate the tests for main effects and interaction, and also 2 tests
   for bacteria type, once just for low temp and once just for high. Use
   contrasts in proc glm.
```

| | BACTERIA TYPE | | |
|------|---------------|------|------|
| TEMP | 1 | 2 | 3 |
| 1 | mu11 | mu12 | mu13 |
| 2 | mu21 | mu22 | mu23 |

Definition: A contrast is a linear combination whose coefficients add to zero. We can test whether collections of contrasts of cell means are all to equal zero. */

```
proc glm;
  title3 'Test contrasts with proc glm';
  class combo;
  model rot=combo;
  means combo / tukey;
  contrast 'Main Effect for Temperature'
    combo 1 1 1 -1 -1 -1;
  contrast 'Main Effect for Bacteria'
    combo 1 -1 0 1 -1 0,
    combo 0 1 -1 0 1 -1;
  contrast 'Temperature by Bacteria Interaction'
    combo 1 -1 0 -1 1 0,
    combo 0 1 -1 0 -1 1;
  contrast 'Bacteria Just for Low Temp'
    combo 1 -1 0 0 0 0,
    combo 0 1 -1 0 0 0;
```

```

contrast 'Bacteria Just for High Temp'
  combo 0 0 0 1 -1 0,
  combo 0 0 0 0 1 -1;
/* Components of the Interaction */
contrast 'Temp Effect: Bact 1 vs 2' /* (mu11-mu21) - (mu12-mu22) */
  combo 1 -1 0 -1 1 0;
contrast 'Temp Effect: Bact 1 vs 3' /* (mu11-mu21) - (mu13-mu23) */
  combo 1 0 -1 -1 0 1;
contrast 'Temp Effect: Bact 2 vs 3' /* (mu12-mu22) - (mu13-mu23) */
  combo 0 1 -1 0 -1 1;
/* One could protect the last 3 tests with a Bonferroni correction as
follow-ups to the significant interaction. Use 0.05/3 = 0.01667 */

```

Rotten potatoes: STA429/1007 F 2007

1

The MEANS Procedure

Analysis Variable : rot

| bact | temp | N Obs | Mean | Std Dev |
|------|------|----------|------------|-----------|
| 1 | 1 | 9 | 3.5555556 | 4.2752518 |
| | 2 | 9 | 7.0000000 | 3.5355339 |
| 2 | 1 | 9 | 4.7777778 | 3.1135903 |
| | 2 | 9 | 13.5555556 | 6.3267510 |
| 3 | 1 | 9 | 8.0000000 | 4.5552168 |
| | 2 | 9 | 19.5555556 | 5.5251948 |

Rotten potatoes: STA429/1007 F 2007

2

| | bact | | | All |
|------|------|-------|-------|-------|
| | 1 | 2 | 3 | |
| | Mean | Mean | Mean | Mean |
| temp | rot | rot | rot | rot |
| 1 | 3.56 | 4.78 | 8.00 | 5.44 |
| 2 | 7.00 | 13.56 | 19.56 | 13.37 |
| All | 5.28 | 9.17 | 13.78 | 9.41 |

Rotten potatoes: STA429/1007 F 2007 3

Standard 2-way ANOVA with proc glm

The GLM Procedure

Class Level Information

| Class | Levels | Values |
|-------|--------|--------|
| bact | 3 | 1 2 3 |
| temp | 2 | 1 2 |

Number of observations 54

Rotten potatoes: STA429/1007 F 2007 4

Standard 2-way ANOVA with proc glm

The GLM Procedure

Dependent Variable: rot

| Source | DF | Sum of Squares | Mean Square | F Value | Pr > F |
|-----------------|----|----------------|-------------|---------|--------|
| Model | 5 | 1652.814815 | 330.562963 | 15.05 | <.0001 |
| Error | 48 | 1054.222222 | 21.962963 | | |
| Corrected Total | 53 | 2707.037037 | | | |

| | | | |
|----------|-----------|----------|----------|
| R-Square | Coeff Var | Root MSE | rot Mean |
| 0.610562 | 49.81676 | 4.686466 | 9.407407 |

| Source | DF | Type I SS | Mean Square | F Value | Pr > F |
|-----------|----|-------------|-------------|---------|--------|
| temp | 1 | 848.0740741 | 848.0740741 | 38.61 | <.0001 |
| bact | 2 | 651.8148148 | 325.9074074 | 14.84 | <.0001 |
| bact*temp | 2 | 152.9259259 | 76.4629630 | 3.48 | 0.0387 |

| Source | DF | Type III SS | Mean Square | F Value | Pr > F |
|-----------|----|-------------|-------------|---------|--------|
| temp | 1 | 848.0740741 | 848.0740741 | 38.61 | <.0001 |
| bact | 2 | 651.8148148 | 325.9074074 | 14.84 | <.0001 |
| bact*temp | 2 | 152.9259259 | 76.4629630 | 3.48 | 0.0387 |

Rotten potatoes: STA429/1007 F 2007

5

Standard 2-way ANOVA with proc glm

The GLM Procedure

| Level of bact | Level of temp | N | -----rot----- Mean | Std Dev |
|------------------|------------------|---|-----------------------|------------|
| 1 | 1 | 9 | 3.5555556 | 4.27525178 |
| 1 | 2 | 9 | 7.0000000 | 3.53553391 |
| 2 | 1 | 9 | 4.7777778 | 3.11359028 |
| 2 | 2 | 9 | 13.5555556 | 6.32675097 |
| 3 | 1 | 9 | 8.0000000 | 4.55521679 |
| 3 | 2 | 9 | 19.5555556 | 5.52519482 |

Rotten potatoes: STA429/1007 F 2007

6

Test contrasts with proc glm

The GLM Procedure

Class Level Information

| Class | Levels | Values |
|-------|--------|-------------------|
| combo | 6 | 11 12 13 21 22 23 |

Number of observations 54

Rotten potatoes: STA429/1007 F 2007

7

Test contrasts with proc glm

The GLM Procedure

Dependent Variable: rot

| Source | DF | Sum of Squares | Mean Square | F Value | Pr > F |
|-----------------|----|-------------------|-------------|---------|--------|
| Model | 5 | 1652.814815 | 330.562963 | 15.05 | <.0001 |
| Error | 48 | 1054.222222 | 21.962963 | | |
| Corrected Total | 53 | 2707.037037 | | | |

| | | | |
|----------|-----------|----------|----------|
| R-Square | Coeff Var | Root MSE | rot Mean |
| 0.610562 | 49.81676 | 4.686466 | 9.407407 |

| Source | DF | Type I SS | Mean Square | F Value | Pr > F |
|--------|----|-------------|-------------|---------|--------|
| combo | 5 | 1652.814815 | 330.562963 | 15.05 | <.0001 |

| Source | DF | Type III SS | Mean Square | F Value | Pr > F |
|--------|----|-------------|-------------|---------|--------|
| combo | 5 | 1652.814815 | 330.562963 | 15.05 | <.0001 |

Rotten potatoes: STA429/1007 F 2007

8

Test contrasts with proc glm

The GLM Procedure

Tukey's Studentized Range (HSD) Test for rot

NOTE: This test controls the Type I experimentwise error rate, but it generally has a higher Type II error rate than REGWQ.

| | |
|-------------------------------------|----------|
| Alpha | 0.05 |
| Error Degrees of Freedom | 48 |
| Error Mean Square | 21.96296 |
| Critical Value of Studentized Range | 4.19724 |
| Minimum Significant Difference | 6.5567 |

Means with the same letter are not significantly different.

| Tukey Grouping | Mean | N | combo |
|----------------|--------|---|-------|
| A | 19.556 | 9 | 23 |
| A | | | |
| B A | 13.556 | 9 | 22 |
| B | | | |
| B C | 8.000 | 9 | 13 |
| B C | | | |
| B C | 7.000 | 9 | 21 |
| C | | | |
| C | 4.778 | 9 | 12 |
| C | | | |
| C | 3.556 | 9 | 11 |

Rotten potatoes: STA429/1007 F 2007

9

Test contrasts with proc glm

The GLM Procedure

Dependent Variable: rot

| Contrast | DF | Contrast SS | Mean Square |
|-------------------------------------|----|-------------|-------------|
| Main Effect for Temperature | 1 | 848.0740741 | 848.0740741 |
| Main Effect for Bacteria | 2 | 651.8148148 | 325.9074074 |
| Temperature by Bacteria Interaction | 2 | 152.9259259 | 76.4629630 |
| Bacteria Just for Low Temp | 2 | 94.8888889 | 47.4444444 |
| Bacteria Just for High Temp | 2 | 709.8518519 | 354.9259259 |
| Temp Effect: Bact 1 vs 2 | 1 | 64.0000000 | 64.0000000 |
| Temp Effect: Bact 1 vs 3 | 1 | 148.0277778 | 148.0277778 |
| Temp Effect: Bact 2 vs 3 | 1 | 17.3611111 | 17.3611111 |

| Contrast | F Value | Pr > F |
|-------------------------------------|---------|--------|
| Main Effect for Temperature | 38.61 | <.0001 |
| Main Effect for Bacteria | 14.84 | <.0001 |
| Temperature by Bacteria Interaction | 3.48 | 0.0387 |
| Bacteria Just for Low Temp | 2.16 | 0.1264 |
| Bacteria Just for High Temp | 16.16 | <.0001 |
| Temp Effect: Bact 1 vs 2 | 2.91 | 0.0943 |
| Temp Effect: Bact 1 vs 3 | 6.74 | 0.0125 |
| Temp Effect: Bact 2 vs 3 | 0.79 | 0.3784 |